'

'

' Excel表Word変換マクロ（e-neh８）第１版H230126

'

'

Public xlApp As Excel.Application

Public TE, CE As Excel.Range

Public ThisMacro As Word.Document

Public TW As Word.Table

Public CW As Word.Cell

Public Spacer As String

Public ScaleFactor As Single

Public NoC, MakeNewDoc As Long

Public Type CPTEntry

    ExlAddr As String

    ExlMerged As Boolean

    EAP As Excel.Range

    WAP As Word.Cell

    WAP1 As Word.Cell

    WAP2 As Word.Cell

End Type

Public CPT() As CPTEntry

Public ScratchPadName As String

'

'（初期設定ルーチン）

'

Sub AutoOpen()

    On Error Resume Next

    Spacer = Chr(vbKeyF16)

    ScaleFactor = 0.92

    Set ThisMacro = ActiveDocument

    '

    ' Wordの作業用ウインドウの表示調整

    '

    MakeNewDoc = MsgBox("新規文書を作成しますか？", vbYesNo, _

                        "e-neh８")

    If MakeNewDoc = vbNo Then

        Dialogs(wdDialogFileOpen).Show

    Else

        Application.Documents.Add

    End If

    With Application

        .WindowState = wdWindowStateNormal

        .Width = Int(.UsableWidth / 2)

        .Height = .UsableHeight

        .Move Left:=Int(.UsableWidth / 2), Top:=0

    End With

    '

    ' 操作用フォームの立上げ

    '

    Load 操作用フォーム

    Do

        操作用フォーム.Show (vbModeless)

    Loop While 操作用フォーム.Visible = False

    '

    ' Excelの立上げと作業用ウインドウの表示調整など

    '

    Set xlApp = CreateObject("Excel.Application")

    DoEvents

    With xlApp

        .Visible = True

        .WindowState = xlNormal

        .Width = Int(Application.UsableWidth / 2)

        .Height = Application.UsableHeight

        .Top = 1

        .Left = 1

    End With

    DoEvents

    xlApp.Dialogs(xlDialogFindFile).Show

End Sub

'

'（Excelファイル変更ルーチン）

'

Public Sub ChangeExcelBook()

    On Error Resume Next

    xlApp.ActiveWorkbook.Close SaveChanges:=False

    xlApp.Visible = True

    DoEvents

    xlApp.Dialogs(xlDialogFindFile).Show

End Sub

'

'（Excel表の基本的な枠組をWord文書へ貼付けルーチン）

'

Public Sub MakeBasicTWFrameAsTE()

    On Error Resume Next

    Dim I As Long

    '

    ' 変換対象シートをコピーした作業用シートの作成

    '

    xlApp.Worksheets(xlApp.ActiveSheet.Name).Copy Before:=ActiveSheet

    ScratchPadName = ActiveSheet.Name

    xlApp.Worksheets(ScratchPadName).Select

    '

    ' 数式の削除及び非表示のカラム及び行の削除

    '

    Set TE = xlApp.Selection

    xlApp.Cells.Select

    xlApp.Selection.Copy

    xlApp.Selection.PasteSpecial Paste:=xlPasteValues

    TE.Select

    For I = TE.Columns.Count To 1 Step -1

        If TE.Columns(I).Hidden Then

            TE.Columns(I).EntireColumn.Delete

        End If

    Next I

    For I = TE.Rows.Count To 1 Step -1

        If TE.Rows(I).Hidden Then

            TE.Rows(I).EntireRow.Delete

        End If

    Next I

    '

    ' Word表の作成

    '

    Set TW = ActiveDocument.Tables.Add( _

            Range:=Selection.Range, _

            NumRows:=TE.Rows.Count, _

            NumColumns:=TE.Columns.Count, _

            DefaultTableBehavior:=wdWord8TableBehavior)

    TW.AllowAutoFit = False

    If MakeNewDoc = vbYes Then

        ActiveDocument.Paragraphs.Add

    End If

    '

    ' セルのアドレス・バッファ（Cell Pointer Table）の再定義

    '

    ReDim CPT(TE.Cells.Count)

    NoC = 0

End Sub

'

'（作業用シートの削除ルーチン）

'

Public Sub DeleteScratchPad()

        Application.DisplayAlerts = False

        xlApp.DisplayAlerts = False

        xlApp.Sheets(ScratchPadName).Delete

        Application.DisplayAlerts = True

        xlApp.DisplayAlerts = True

End Sub

'

'（列幅及び行高さの調整ルーチン）

'

Public Sub AdjustColumnsAndRows()

    Dim I As Long

    On Error Resume Next

    '

    ' 行グリッドからの解放及び余白のクリア

    '

    TW.Range.ParagraphFormat.DisableLineHeightGrid = True

    TW.LeftPadding = 2

    TW.Range.ParagraphFormat.LeftIndent = 0

    TW.RightPadding = 2

    TW.Range.ParagraphFormat.RightIndent = 0

    TW.TopPadding = 1

    TW.BottomPadding = 0

    '

    ' 列幅及び行高さの調整

    '

    TW.Cell(1, 1).Select    ' monitor

    For I = 1 To TE.Columns.Count

        TW.Columns(I).Select    ' monitor

        DoEvents                ' monitor

        If TE.Columns(I).Width * ScaleFactor < 10.5 Then

            TW.Columns(I).Width = 10.5

        Else

            TW.Columns(I).Width = TE.Columns(I).Width * ScaleFactor

        End If

    Next I

    For I = 1 To TE.Rows.Count

        TW.Rows(I).Select       'monitor

        DoEvents                'monitor

        TW.Rows(I).HeightRule = wdRowHeightExactly

        TW.Rows(I).Height = TE.Rows(I).Height * ScaleFactor

    Next I

End Sub

'

'（Word表の各セルをExcel表と同じように結合ルーチン）

'

Public Sub MergeCellsOfTW()

    Dim C As Excel.Range

    Dim AddressStack, TopLeft, CellAddress, CellAddress2 As String

    Dim TLRowBase, TLColumnBase As Long

    Dim I, J, K, L As Long

    Dim TLRow, TLColumn As Long

    Dim BRRow, BRColumn As Long

    On Error Resume Next

    '

    ' 結合セルの洗出し

    '

    AddressStack = Spacer

    TopLeft = TE.Address(ReferenceStyle:=xlR1C1) & ":"

    I = InStr(1, TopLeft, "C")

    TLRowBase = Val(Mid(Left(TopLeft, I - 1), 2)) - 1

    J = InStr(1, TopLeft, ":")

    TLColumnBase = Val(Mid(Left(TopLeft, J - 1), I + 1)) - 1

    For K = 1 To TE.Rows.Count

        For L = 1 To TE.Columns.Count

            Set C = TE.Cells(K, L)

            If C.MergeCells Then

                CellAddress = C.MergeArea.Address(ReferenceStyle:=xlR1C1)

                CellAddress2 = C.MergeArea.Address(ReferenceStyle:=xlA1)

                If InStr(1, AddressStack, Spacer & CellAddress & Spacer) = 0 Then

                    '

                    ' 未登録結合セル

                    '

                    NoC = NoC + 1

                    TopLeft = CellAddress & ":"

                    I = InStr(1, TopLeft, "C")

                    TLRow = Val(Mid(Left(TopLeft, I - 1), 2)) - TLRowBase

                    J = InStr(1, TopLeft, ":")

                    TLColumn = Val(Mid(Left(TopLeft, J - 1), I + 1)) - TLColumnBase

                    I = InStr(J + 1, TopLeft, "C")

                    BRRow = Val(Mid(Left(TopLeft, I - 1), J + 2)) - TLRowBase

                    J = InStr(J + 1, TopLeft, ":")

                    BRColumn = Val(Mid(Left(TopLeft, J - 1), I + 1)) - TLColumnBase

                    With CPT(NoC)

                        .ExlAddr = CellAddress2

                        .ExlMerged = True

                        Set .EAP = xlApp.ActiveSheet.Range(CellAddress2)

                        Set .WAP1 = TW.Cell(TLRow, TLColumn)

                        Set .WAP2 = TW.Cell(BRRow, BRColumn)

                    End With

                    '

                    ' 結合セルの処理済み登録

                    '

                    AddressStack = AddressStack & CellAddress & Spacer

                End If

            Else

                NoC = NoC + 1

                CellAddress = C.Address(ReferenceStyle:=xlR1C1)

                CellAddress2 = C.Address(ReferenceStyle:=xlA1)

                TopLeft = CellAddress & ":"

                I = InStr(1, TopLeft, "C")

                TLRow = Val(Mid(Left(TopLeft, I - 1), 2)) - TLRowBase

                J = InStr(1, TopLeft, ":")

                TLColumn = Val(Mid(Left(TopLeft, J - 1), I + 1)) - TLColumnBase

                With CPT(NoC)

                    .ExlAddr = CellAddress2

                    .ExlMerged = False

                    Set .EAP = xlApp.ActiveSheet.Range(CellAddress2)

                    Set .WAP = TW.Cell(TLRow, TLColumn)

                End With

            End If

        Next L

    Next K

    '

    ' セルの一括結合

    '

    For I = 1 To NoC

        With CPT(I)

            If .ExlMerged Then

                .WAP1.Merge Mergeto:=.WAP2

                Set .WAP = .WAP1

                .WAP.Select     ' monitor

                DoEvents        ' monitor

            End If

        End With

    Next I

End Sub

'

'（セル内の配置の調整ルーチン）

'

Public Sub AdjustCellFormat()

    Dim I As Long

    Dim HeightToRecover As Long

    On Error Resume Next

    For I = 1 To NoC

        Set CE = CPT(I).EAP

        Set CW = CPT(I).WAP

        CW.Select   ' monitor

        DoEvents    ' monitor

        '

        ' 縦書きの調整

        '

        If CE.Orientation = xlVertical Then

            HeightToRecover = CW.Height

            CW.Range.Orientation = wdTextOrientationVerticalFarEast

            CW.Height = HeightToRecover

        End If

        '

        ' 横書き／縦書きに応じた配置の調整

        '

        If CE.Orientation = xlHorizontal Then

            '

            ' 横書きにおける縦方向の配置の調整

            '

            Select Case CE.VerticalAlignment

                Case xlCenter

                    CW.VerticalAlignment = wdCellAlignVerticalCenter

                Case xlBottom

                    CW.VerticalAlignment = wdCellAlignVerticalBottom

                Case xlTop

                    CW.VerticalAlignment = wdCellAlignVerticalTop

                Case xlJustify

                    CW.VerticalAlignment = wdCellAlignVerticalTop

                Case xlDistributed

                    CW.VerticalAlignment = wdCellAlignVerticalCenter

                Case Else

                    CW.VerticalAlignment = wdCellAlignVerticalCenter

            End Select

            '

            ' 横書きにおける水平方向の配置の調整

            '

            With CW.Range.ParagraphFormat

                Select Case CE.HorizontalAlignment

                    Case xlCenter

                        .Alignment = wdAlignParagraphCenter

                    Case xlDistributed

                        .Alignment = wdAlignParagraphDistribute

                        .CharacterUnitLeftIndent = CE.IndentLevel

                        .CharacterUnitFirstLineIndent = 0

                        .CharacterUnitRightIndent = CE.IndentLevel

                    Case xlHAlignGeneral

                        If IsNumber(CE.Range("A1").Text) Then

                            .Alignment = wdAlignParagraphRight

                            .CharacterUnitRightIndent = 0.5

                        ElseIf IsPercentage(CE.Range("A1").Text) Then

                            .Alignment = wdAlignParagraphRight

                            .CharacterUnitRightIndent = 0

                        Else

                            .Alignment = wdAlignParagraphLeft

                            .CharacterUnitLeftIndent = 0

                            .CharacterUnitFirstLineIndent = 0

                        End If

                    Case xlJustify

                        .Alignment = wdAlignParagraphJustify

                    Case xlLeft

                        .Alignment = wdAlignParagraphLeft

                        .CharacterUnitLeftIndent = CE.IndentLevel

                        .CharacterUnitFirstLineIndent = 0

                    Case xlRight

                        .Alignment = wdAlignParagraphRight

                        .CharacterUnitRightIndent = CE.IndentLevel

                    Case Else

                        .Alignment = wdAlignParagraphCenter

                End Select

            End With

        Else

            '

            ' 縦書きにおける縦方向の配置の調整

            '

            With CW.Range.ParagraphFormat

                .LeftIndent = 0

                .RightIndent = 0

                Select Case CE.VerticalAlignment

                    Case xlCenter

                        .Alignment = wdAlignParagraphCenter

                    Case xlJustify

                        .Alignment = wdAlignParagraphJustify

                    Case xlBottom

                        .Alignment = wdAlignParagraphRight

                        .CharacterUnitRightIndent = CE.IndentLevel

                    Case xlDistributed

                        .Alignment = wdAlignParagraphDistribute

                        .CharacterUnitLeftIndent = CE.IndentLevel

                        .CharacterUnitFirstLineIndent = 0

                        .CharacterUnitRightIndent = CE.IndentLevel

                    Case xlTop

                        .Alignment = wdAlignParagraphLeft

                        .CharacterUnitLeftIndent = CE.IndentLevel

                        .CharacterUnitFirstLineIndent = 0

                    Case Else

                        .Alignment = wdAlignParagraphCenter

                End Select

            End With

            '

            ' 縦書きにおける水平方向の配置の調整

            '

            Select Case CE.HorizontalAlignment

                Case xlCenter

                    CW.VerticalAlignment = wdCellAlignVerticalCenter

                Case xlLeft

                    CW.VerticalAlignment = wdCellAlignVerticalTop

                Case xlRight

                    CW.VerticalAlignment = wdCellAlignVerticalBottom

                Case xlJustify

                    CW.VerticalAlignment = wdCellAlignVerticalBottom

                Case xlDistributed

                    CW.VerticalAlignment = wdCellAlignVerticalCenter

                Case Else

                    CW.VerticalAlignment = wdCellAlignVerticalCenter

            End Select

        End If

        '

        ' 文字表示方法の調整

        '

        With CW.Range.ParagraphFormat

            .LineSpacingRule = wdLineSpaceMultiple

            .LineSpacing = LinesToPoints(0.85)

        End With

        CW.WordWrap = CE.Range("A1").WrapText

        If Not CW.WordWrap Then

            CW.FitText = CE.Range("A1").ShrinkToFit

        End If

        '

        ' データのテキストとしてのコピー

        '

        CW.Range.Text = CE.Range("A1").Text

        If Len(CE.Range("A1").Text) = 0 Then

            CW.Range.Font.Name = CE.Range("A1").Font.Name

            CW.Range.Font.Size = CE.Range("A1").Font.Size

        Else

            CW.Range.Font.Name = CE.Range("A1").Characters(1, 1).Font.Name

            CW.Range.Font.Size = CE.Range("A1").Characters(1, 1).Font.Size

        End If

        '

        ' 縦書きにおけるFitText向け調整

        '

        If (CW.Range.Orientation = wdTextOrientationVerticalFarEast) And _

                CW.FitText Then

            CW.Range.ParagraphFormat.Alignment = wdAlignParagraphCenter

            CW.RightPadding = 2

            CW.LeftPadding = 2

            CW.TopPadding = 0

            CW.BottomPadding = 0

            CW.Range.FitTextWidth = CE.Height * ScaleFactor

        End If

    Next I

End Sub

'

'（数値表示判定ルーチン）

'

Private Function IsNumber(ByVal S As String) As Boolean

    Select Case Left(S, 1)

        Case "-", "－"

            IsNumber = IsNumeric(Mid(S, 2)) Or (Mid(S, 2) = "")

        Case "△", "▲"

            IsNumber = IsNumeric(Mid(S, 2))

        Case Else

            IsNumber = IsNumeric(S)

    End Select

End Function

'

'（パーセント表示判定ルーチン）

'

Private Function IsPercentage(ByVal S As String) As Boolean

    IsPercentage = IsNumber(Left(S, Len(S) - 1)) _

                    And (Right(S, 1) = "%" Or Right(S, 1) = "％")

End Function

'

'（セル罫線の一括調整ルーチン）

'

Public Sub AdjustEachCellBorder()

    Dim I As Long

    On Error Resume Next

    For I = 1 To NoC

        Set CE = CPT(I).EAP

        Set CW = CPT(I).WAP

        CW.Select   ' monitor

        DoEvents    ' monitor

        Call AdjustBorder(wdBorderTop, xlEdgeTop)

        Call AdjustBorder(wdBorderLeft, xlEdgeLeft)

        Call AdjustBorder(wdBorderRight, xlEdgeRight)

        Call AdjustBorder(wdBorderBottom, xlEdgeBottom)

        Call AdjustBorder(wdBorderDiagonalDown, xlDiagonalDown)

        Call AdjustBorder(wdBorderDiagonalUp, xlDiagonalUp)

    Next I

    '

    ' 整合性チェック

    '

    For I = 1 To NoC

        Set CE = CPT(I).EAP

        Set CW = CPT(I).WAP

        CW.Select   ' monitor

        DoEvents    ' monitor

        Call CheckBorder(wdBorderTop, xlEdgeTop)

        Call CheckBorder(wdBorderLeft, xlEdgeLeft)

        Call CheckBorder(wdBorderRight, xlEdgeRight)

        Call CheckBorder(wdBorderBottom, xlEdgeBottom)

        Call CheckBorder(wdBorderDiagonalDown, xlDiagonalDown)

        Call CheckBorder(wdBorderDiagonalUp, xlDiagonalUp)

    Next I

    '

    ' 後始末

    '

    TE.Select

    If MakeNewDoc = vbYes Then

        ActiveDocument.Paragraphs.Last.Range.Select

    Else

        TW.Range.Paragraphs.Last.Range.Select

        Selection.Move Unit:=wdParagraph, Count:=1

    End If

End Sub

'

'（セル罫線の個別調整ルーチン）

'

Private Sub AdjustBorder(ByVal WB, EB As Long)

    On Error Resume Next

    If CE.Borders(EB).LineStyle = xlLineStyleNone Then Exit Sub

    If MultiLineStyle(EB) Then Exit Sub

    Select Case CE.Borders(EB).LineStyle

        Case xlContinuous

            CW.Borders(WB).LineStyle = wdLineStyleSingle

        Case xlDash

            CW.Borders(WB).LineStyle = wdLineStyleDashSmallGap

        Case xlDashDot

            CW.Borders(WB).LineStyle = wdLineStyleDashDot

        Case xlDashDotDot

            CW.Borders(WB).LineStyle = wdLineStyleDashDotDot

        Case xlDot

            CW.Borders(WB).LineStyle = wdLineStyleDot

        Case xlDouble

            CW.Borders(WB).LineStyle = wdLineStyleDouble

            CW.Borders(WB).LineWidth = wdLineWidth075pt

            Exit Sub

        Case xlSlantDashDot

            CW.Borders(WB).LineStyle = wdLineStyleDashDotStroked

            CW.Borders(WB).LineWidth = wdLineWidth300pt

            Exit Sub

        Case Else

    End Select

    Select Case CE.Borders(EB).Weight

        Case xlHairline

            CW.Borders(WB).LineWidth = wdLineWidth050pt

        Case xlMedium

            CW.Borders(WB).LineWidth = wdLineWidth150pt

        Case xlThick

            CW.Borders(WB).LineWidth = wdLineWidth300pt

        Case xlThin

            CW.Borders(WB).LineWidth = wdLineWidth100pt

        Case Else

    End Select

End Sub

'

'（線種が多種類か判定ルーチン）

'

Private Function MultiLineStyle(ByVal EB As Long) As Boolean

    Dim FirstLineStyle, FirstLineWeight, I As Long

    Dim R, C As Excel.Range

    On Error Resume Next

    MultiLineStyle = False

    Select Case EB

        Case xlEdgeTop

            Set R = CE.Rows(1)

        Case xlEdgeLeft

            Set R = CE.Columns(1)

        Case xlEdgeRight

            Set R = CE.Columns(CE.Columns.Count)

        Case xlEdgeBottom

            Set R = CE.Rows(CE.Rows.Count)

        Case xlDiagonalDown

            Exit Function

        Case xlDiagonalUp

            Exit Function

        Case Else

            Exit Function

    End Select

    FirstLineStyle = R.Cells(1).Borders(EB).LineStyle

    FirstLineWeight = R.Cells(1).Borders(EB).Weight

    For I = 1 To R.Cells.Count

        If R.Cells(I).Borders(EB).LineStyle <> FirstLineStyle Or _

                R.Cells(I).Borders(EB).Weight <> FirstLineWeight Then

            MultiLineStyle = True

            Exit Function

        End If

    Next I

End Function

'

'（罫線の整合性チェックルーチン）

'

Private Sub CheckBorder(ByVal WB, EB As Long)

    On Error Resume Next

    If MultiLineStyle(EB) Then Exit Sub

    Select Case CE.Borders(EB).LineStyle

        Case xlLineStyleNone

            If CW.Borders(WB).LineStyle <> wdLineStyleNone Then

                CW.Borders(WB).LineStyle = wdLineStyleNone

            End If

        Case xlContinuous

            If CW.Borders(WB).LineStyle <> wdLineStyleSingle Then

                CW.Borders(WB).LineStyle = wdLineStyleSingle

            End If

        Case xlDash

            If CW.Borders(WB).LineStyle <> wdLineStyleDashSmallGap Then

                CW.Borders(WB).LineStyle = wdLineStyleDashSmallGap

            End If

        Case xlDashDot

            If CW.Borders(WB).LineStyle <> wdLineStyleDashDot Then

                CW.Borders(WB).LineStyle = wdLineStyleDashDot

            End If

        Case xlDashDotDot

            If CW.Borders(WB).LineStyle <> wdLineStyleDashDotDot Then

                CW.Borders(WB).LineStyle = wdLineStyleDashDotDot

            End If

        Case xlDot

            If CW.Borders(WB).LineStyle <> wdLineStyleDot Then

                CW.Borders(WB).LineStyle = wdLineStyleDot

            End If

        Case xlDouble

            If CW.Borders(WB).LineStyle <> wdLineStyleDouble Then

                CW.Borders(WB).LineStyle = wdLineStyleDouble

                CW.Borders(WB).LineWidth = wdLineWidth075pt

                Exit Sub

            Else

                Exit Sub

            End If

        Case xlSlantDashDot

            If CW.Borders(WB).LineStyle <> wdLineStyleDashDotStroked Then

                CW.Borders(WB).LineStyle = wdLineStyleDashDotStroked

                CW.Borders(WB).LineWidth = wdLineWidth300pt

                Exit Sub

            Else

                Exit Sub

            End If

        Case Else

    End Select

    Select Case CE.Borders(EB).Weight

        Case xlHairline

            If CW.Borders(WB).LineWidth <> wdLineWidth050pt Then

                CW.Borders(WB).LineWidth = wdLineWidth050pt

            End If

        Case xlMedium

            If CW.Borders(WB).LineWidth <> wdLineWidth150pt Then

                CW.Borders(WB).LineWidth = wdLineWidth150pt

            End If

        Case xlThick

            If CW.Borders(WB).LineWidth <> wdLineWidth300pt Then

                CW.Borders(WB).LineWidth = wdLineWidth300pt

            End If

        Case xlThin

            If CW.Borders(WB).LineWidth <> wdLineWidth100pt Then

                CW.Borders(WB).LineWidth = wdLineWidth100pt

            End If

        Case Else

    End Select

End Sub

'

'（終了処理ルーチン）

'

Public Sub ClosingProcess()

    On Error Resume Next

    xlApp.WindowState = xlMaximized

    xlApp.ActiveWorkbook.Close SaveChanges:=False

    xlApp.Visible = False

    xlApp.Quit

    Set xlApp = Nothing

    Application.WindowState = wdWindowStateMaximize

    DoEvents

    ThisMacro.Close

End Sub

'

' 縦貼付け用ルーチン群

'

'（Excel表の基本的な枠組をWord文書へ貼付けルーチン）

'

Public Sub MakeBasicTWFrameAsTE2()

    On Error Resume Next

    Dim I As Long

    '

    ' 変換対象シートをコピーした作業用シートの作成

    '

    xlApp.Worksheets(xlApp.ActiveSheet.Name).Copy Before:=ActiveSheet

    ScratchPadName = ActiveSheet.Name

    xlApp.Worksheets(ScratchPadName).Select

    '

    ' 数式の削除及び非表示のカラム及び行の削除

    Set TE = xlApp.Selection

    xlApp.Cells.Select

    xlApp.Selection.Copy

    xlApp.Selection.PasteSpecial pasete:=xlPasteValues

    TE.Select

    For I = TE.Columns.Count To 1 Step -1

        If TE.Columns(I).Hidden Then

            TE.Columns(I).entirecolumns.Delete

        End If

    Next I

    For I = TE.Rows.Count To 1 Step -1

        If TE.Rows(I).Hidden Then

            TE.Rows(I).EntireRow.Delete

        End If

    Next I

    '

    ' Word表の作成など

    '

    Set TW = ActiveDocument.Tables.Add( _

            Range:=Selection.Range, _

            NumRows:=TE.Columns.Count, _

            NumColumns:=TE.Rows.Count, _

            DefaultTableBehavior:=wdWord8TableBehavior)

    TW.AllowAutoFit = False

    If MakeNewDoc = vbYes Then

        ActiveDocument.Paragraphs.Add

    End If

    '

    ' セルのアドレス・バッファ（Cell Pointer Table）の再定義

    '

    ReDim CPT(TE.Cells.Count)

    NoC = 0

End Sub

'

'（列幅及び行高さの調整ルーチン）

'

Public Sub AdjustColumnsAndRows2()

    Dim I As Long

    On Error Resume Next

    '

    ' 行グリッドからの解放及び余白の調整

    '

    TW.Range.ParagraphFormat.DisableLineHeightGrid = True

    TW.LeftPadding = 2

    TW.Range.ParagraphFormat.LeftIndent = 0

    TW.RightPadding = 2

    TW.Range.ParagraphFormat.RightIndent = 0

    TW.TopPadding = 0

    TW.BottomPadding = 0

    '

    ' 列幅及び行高さの調整

    '

    TW.Cell(1, 1).Select        ' monitor

    For I = 1 To TE.Rows.Count

        TW.Columns(I).Select    ' monitor

        DoEvents                ' monitor

        If TE.Rows(I).Height * ScaleFactor < 10.5 Then

            TW.Columns(I).Width = 10.5

        Else

            TW.Columns(I).Width = TE.Rows(I).Height * ScaleFactor

        End If

    Next I

    For I = TE.Columns.Count To 1 Step -1

        TW.Rows(TE.Columns.Count - I + 1).Select    ' monitor

        DoEvents                                    ' monitor

        TW.Rows(TE.Columns.Count - I + 1).HeightRule = wdRowHeightExactly

        TW.Rows(TE.Columns.Count - I + 1).Height _

            = TE.Columns(I).Width * ScaleFactor

    Next I

End Sub

'

'（Word表の各セルをExcel表と同じように結合ルーチン）

'

Public Sub MergeCellsOfTW2()

    Dim C As Excel.Range

    Dim AddressStack, TopLeft, CellAddress, CellAddress2 As String

    Dim TLRowBase, TLColumnBase As Long

    Dim I, J, K, L As Long

    Dim TLRow, TLColumn As Long

    Dim BRRow, BRColumn As Long

    On Error Resume Next

    '

    ' 結合セルの洗出し

    '

    AddressStack = Spacer

    TopLeft = TE.Address(ReferenceStyle:=xlR1C1) & ":"

    I = InStr(1, TopLeft, "C")

    TLRowBase = Val(Mid(Left(TopLeft, I - 1), 2)) - 1

    J = InStr(1, TopLeft, ":")

    TLColumnBase = Val(Mid(Left(TopLeft, J - 1), I + 1)) - 1

    For K = 1 To TE.Rows.Count

        For L = 1 To TE.Columns.Count

            Set C = TE.Cells(K, L)

            If C.MergeCells Then

                CellAddress = C.MergeArea.Address(ReferenceStyle:=xlR1C1)

                CellAddress2 = C.MergeArea.Address(ReferenceStyle:=xlA1)

                If InStr(1, AddressStack, Spacer & CellAddress & Spacer) = 0 Then

                    '

                    ' 未登録結合セル

                    '

                    NoC = NoC + 1

                    TopLeft = CellAddress & ":"

                    I = InStr(1, TopLeft, "C")

                    TLRow = Val(Mid(Left(TopLeft, I - 1), 2)) - TLRowBase

                    J = InStr(1, TopLeft, ":")

                    TLColumn = Val(Mid(Left(TopLeft, J - 1), I + 1)) - TLColumnBase

                    I = InStr(J + 1, TopLeft, "C")

                    BRRow = Val(Mid(Left(TopLeft, I - 1), J + 2)) - TLRowBase

                    J = InStr(J + 1, TopLeft, ":")

                    BRColumn = Val(Mid(Left(TopLeft, J - 1), I + 1)) - TLColumnBase

                    With CPT(NoC)

                        .ExlAddr = CellAddress2

                        .ExlMerged = True

                        Set .EAP = xlApp.ActiveSheet.Range(CellAddress2)

                        Set .WAP1 = TW.Cell(TE.Columns.Count - BRColumn + 1, TLRow)

                        Set .WAP2 = TW.Cell(TE.Columns.Count - TLColumn + 1, BRRow)

                    End With

                    '

                    ' 結合セルの処理済み登録

                    '

                    AddressStack = AddressStack & CellAddress & Spacer

                End If

            Else

                NoC = NoC + 1

                CellAddress = C.Address(ReferenceStyle:=xlR1C1)

                CellAddress2 = C.Address(ReferenceStyle:=xlA1)

                TopLeft = CellAddress & ":"

                I = InStr(1, TopLeft, "C")

                TLRow = Val(Mid(Left(TopLeft, I - 1), 2)) - TLRowBase

                J = InStr(1, TopLeft, ":")

                TLColumn = Val(Mid(Left(TopLeft, J - 1), I + 1)) - TLColumnBase

                With CPT(NoC)

                    .ExlAddr = CellAddress2

                    .ExlMerged = False

                    Set .EAP = xlApp.ActiveSheet.Range(CellAddress2)

                    Set .WAP = TW.Cell(TE.Columns.Count - TLColumn + 1, TLRow)

                End With

            End If

        Next L

    Next K

    '

    ' セルの一括結合

    '

    For I = 1 To NoC

        With CPT(I)

            If .ExlMerged Then

                .WAP1.Merge Mergeto:=.WAP2

                Set .WAP = .WAP1

                .WAP.Select     ' monitor

                DoEvents        ' monitor

            End If

        End With

    Next I

End Sub

'

'（セル内の配置の調整ルーチン）

'

Public Sub AdjustCellFormat2()

    Dim I As Long

    Dim HeightToRecover As Long

    On Error Resume Next

    For I = 1 To NoC

        Set CE = CPT(I).EAP

        Set CW = CPT(I).WAP

        CW.Select   ' monitor

        DoEvents    ' monitor

        '

        ' 縦書きの調整

        '

        If CE.Orientation = xlVertical Then

            CW.Range.Orientation = wdTextOrientationHorizontalRotatedFarEast

        Else

            HeightToRecover = CW.Height

            CW.Range.Orientation = wdTextOrientationUpward

            CW.Height = HeightToRecover

        End If

        '

        ' 横書き／縦書きに応じた配置の調整

        '

        If CE.Orientation = xlHorizontal Then

            '

            ' 横書き縦貼付けにおける縦方向の配置の調整

            '

            Select Case CE.VerticalAlignment

                Case xlCenter

                    CW.VerticalAlignment = wdCellAlignVerticalCenter

                Case xlBottom

                    CW.VerticalAlignment = wdCellAlignVerticalBottom

                Case xlTop

                    CW.VerticalAlignment = wdCellAlignVerticalTop

                Case xlJustify

                    CW.VerticalAlignment = wdCellAlignVerticalTop

                Case xlDistributed

                    CW.VerticalAlignment = wdCellAlignVerticalCenter

                Case Else

                    CW.VerticalAlignment = wdCellAlignVerticalCenter

            End Select

            '

            ' 横書き縦貼付けにおける水平方向の配置の調整

            '

            With CW.Range.ParagraphFormat

                .LeftIndent = 0

                .RightIndent = 0

                Select Case CE.HorizontalAlignment

                    Case xlCenter

                        .Alignment = wdAlignParagraphCenter

                    Case xlDistributed

                        .Alignment = wdAlignParagraphDistribute

                        .CharacterUnitLeftIndent = CE.IndentLevel

                        .CharacterUnitFirstLineIndent = 0

                        .CharacterUnitRightIndent = CE.IndentLevel

                    Case xlHAlignGeneral

                        If IsNumber(CE.Range("A1").Text) Then

                            .Alignment = wdAlignParagraphRight

                            .CharacterUnitRightIndent = 0.5

                        ElseIf IsPercentage(CE.Range("A1").Text) Then

                            .Alignment = wdAlignParagraphRight

                            .CharacterUnitRightIndent = 0

                        Else

                            .Alignment = wdAlignParagraphLeft

                            .CharacterUnitLeftIndent = CE.IndentLevel

                            .CharacterUnitFirstLineIndent = 0

                        End If

                    Case xlJustify

                        .Alignment = wdAlignParagraphJustify

                    Case xlLeft

                        .Alignment = wdAlignParagraphLeft

                        .CharacterUnitLeftIndent = CE.IndentLevel

                        .CharacterUnitFirstLineIndent = 0

                    Case xlRight

                        .Alignment = wdAlignParagraphRight

                        .CharacterUnitRightIndent = CE.IndentLevel

                    Case Else

                        .Alignment = wdAlignParagraphCenter

                End Select

            End With

        Else

            '

            ' 縦書き縦貼付けにおける縦方向の配置の調整

            '

            With CW.Range.ParagraphFormat

                Select Case CE.VerticalAlignment

                    Case xlCenter

                        .Alignment = wdAlignParagraphCenter

                    Case xlJustify

                        .Alignment = wdAlignParagraphJustify

                    Case xlBottom

                        .Alignment = wdAlignParagraphRight

                        .CharacterUnitRightIndent = CE.IndentLevel

                    Case xlDistributed

                        .Alignment = wdAlignParagraphDistribute

                        .CharacterUnitLeftIndent = CE.IndentLevel

                        .CharacterUnitFirstLineIndent = 0

                        .CharacterUnitRightIndent = CE.IndentLevel

                    Case xlTop

                        .Alignment = wdAlignParagraphLeft

                        .CharacterUnitLeftIndent = CE.IndentLevel

                        .CharacterUnitFirstLineIndent = 0

                    Case Else

                        .Alignment = wdAlignParagraphCenter

                End Select

            End With

            '

            ' 縦書き縦貼付けにおける水平方向の配置の調整

            '

            Select Case CE.HorizontalAlignment

                Case xlCenter

                    CW.VerticalAlignment = wdCellAlignVerticalCenter

                Case xlLeft

                    CW.VerticalAlignment = wdCellAlignVerticalBottom

                Case xlRight

                    CW.VerticalAlignment = wdCellAlignVerticalTop

                Case xlJustify

                    CW.VerticalAlignment = wdCellAlignVerticalTop

                Case xlDistributed

                    CW.VerticalAlignment = wdCellAlignVerticalCenter

                Case Else

                    CW.VerticalAlignment = wdCellAlignVerticalCenter

            End Select

        End If

        '

        ' 文字表示方法の調整

        '

        With CW.Range.ParagraphFormat

            .LineSpacingRule = wdLineSpaceMultiple

            .LineSpacing = LinesToPoints(0.9)

        End With

        CW.WordWrap = CE.Range("A1").WrapText

        If Not CW.WordWrap Then

            CW.FitText = CE.Range("A1").ShrinkToFit

        End If

        '

        ' データのテキストとしてのコピー

        '

        CW.Range.Text = CE.Range("A1").Text

        If Len(CE.Range("A1").Text) = 0 Then

            CW.Range.Font.Name = CE.Range("A1").Font.Name

            CW.Range.Font.Size = CE.Range("A1").Font.Size

        Else

            CW.Range.Font.Name = CE.Range("A1").Characters(1, 1).Font.Name

            CW.Range.Font.Size = CE.Range("A1").Characters(1, 1).Font.Size

        End If

        If (CW.Range.Orientation = wdTextOrientationUpward) And _

                CW.FitText Then

            CW.Range.ParagraphFormat.Alignment = wdAlignParagraphCenter

            CW.RightPadding = 0

            CW.LeftPadding = 0

            CW.TopPadding = 0

            CW.BottomPadding = 0

            CW.Range.FitTextWidth = CE.Width * ScaleFactor

        End If

    Next I

End Sub

'

'（セル罫線の一括調整ルーチン）

'

Public Sub AdjustEachCellBorder2()

    Dim I As Long

    On Error Resume Next

    For I = 1 To NoC

        Set CE = CPT(I).EAP

        Set CW = CPT(I).WAP

        CW.Select   ' monitor

        DoEvents    ' monitor

        Call AdjustBorder(wdBorderLeft, xlEdgeTop)

        Call AdjustBorder(wdBorderBottom, xlEdgeLeft)

        Call AdjustBorder(wdBorderTop, xlEdgeRight)

        Call AdjustBorder(wdBorderRight, xlEdgeBottom)

        Call AdjustBorder(wdBorderDiagonalUp, xlDiagonalDown)

        Call AdjustBorder(wdBorderDiagonalDown, xlDiagonalUp)

    Next I

    '

    ' 整合性チェック

    '

    For I = 1 To NoC

        Set CE = CPT(I).EAP

        Set CW = CPT(I).WAP

        CW.Select   ' monitor

        DoEvents    ' monitor

        Call CheckBorder(wdBorderLeft, xlEdgeTop)

        Call CheckBorder(wdBorderBottom, xlEdgeLeft)

        Call CheckBorder(wdBorderTop, xlEdgeRight)

        Call CheckBorder(wdBorderRight, xlEdgeBottom)

        Call CheckBorder(wdBorderDiagonalUp, xlDiagonalDown)

        Call CheckBorder(wdBorderDiagonalDown, xlDiagonalUp)

        DoEvents

    Next I

    '

    ' 後始末

    '

    TE.Select

    If MakeNewDoc = vbYes Then

        ActiveDocument.Paragraphs.Last.Range.Select

    Else

        TW.Range.Paragraphs.Last.Range.Select

        Selection.Move Unit:=wdParagraph, Count:=1

    End If

End Sub

'

' 操作用フォームのイベント・プロシージャ

'

Private Sub 貼付けボタン_Click()

    On Error Resume Next

    Call MakeBasicTWFrameAsTE

    Call AdjustColumnsAndRows

    Call MergeCellsOfTW

    Call AdjustCellFormat

    Call AdjustEachCellBorder

    Call DeleteScratchPad

End Sub

'

Private Sub 縦貼付けボタン_Click()

    On Error Resume Next

    Call MakeBasicTWFrameAsTE2

    Call AdjustColumnsAndRows2

    Call MergeCellsOfTW2

    Call AdjustCellFormat2

    Call AdjustEachCellBorder2

    Call DeleteScratchPad

End Sub

'

Private Sub ファイル変更ボタン_Click()

    On Error Resume Next

    Call ChangeExcelBook

End Sub

'

Private Sub UserForm_QueryClose(Cancel As Integer, CloseMode As Integer)

    On Error Resume Next

    Cancel = 1

    Call ClosingProcess

    Cancel = 0

    Unload Me

End Sub
